13-Aarhus Day 0 - Pre class: Kauffman, Jones, Alexander, Khovanov

May-25-13
3:41 AM

Pre-plan:

Mon May 27, pre-class. The Kauffman bracket and the Jones polynomial (with computations!), the Alexander polynomial, Khovanov homology.

Pasted from <http://www.math.toronto.edu/~drorbn/Talks/Aarhus-1305/#plan>

Plan: 1. The Kauffman bracket & Jones polynomial, to invariance & $X_+ - X_- =)($

2. First program

3. Tangles & 2nd program (as in KnotTheory` manual)

4. The Alexander polynomial as in Sandbjerg.





5. Very basic Khovanov homology [knot homology is not my goal, it is just an entertaining thing to keep in mind]

On board:

This is not the "master class" of http://www.math.toronto.edu/~drorbn/Talks/Aarhus-1305/

Today: Kauffman, Jones, Alex., Khovanov



[KBDef] $\langle\emptyset\rangle = 1; \qquad \langle \bigcirc L\rangle = (-A^2 - B^2)\langle L\rangle; \qquad \langle \times \rangle = A\langle \asymp \rangle + B\langle)(\rangle;$

$$J(L) = (-A^3)^{w(L)} \left.\frac{\langle L\rangle}{\langle \bigcirc \rangle}\right|_{A\to q^{1/4}},$$

In[1]:=
```
<< KnotTheory`
```

Loading KnotTheory` version of February 5, 2013, 3:48:46.4762.
Read more at http://katlas.org/wiki/KnotTheory.

In[2]:=
```
TubePlot[TorusKnot[5, 4]]
```

Out[2]=

In[3]:=
```
SetAttributes[P, Orderless];
KB0[pd_] := Expand[
  Expand[Times @@ pd /. X[a_, b_, c_, d_] ⧴ A P[a, d] P[b, c] + 1 / A P[a, b] P[c, d]] //.
   {P[a_, b_] P[b_, c_] ⧴ P[a, c], P[a_, b_]^2 ⧴ P[a, a], P[a_, a_] → -A^2 - 1 / A^2}
 ]
```

In[5]:=
```
TorusKnot[5, 4] // PD // KB0 // Timing
```

Out[5]= $\left\{17.971315, \frac{1}{A^{23}} + \frac{1}{A^{19}} + \frac{1}{A^{15}} + \frac{1}{A^{11}} + \frac{1}{A^{7}} - A - A^{5} - A^{9}\right\}$

In[6]:=
```
SetAttributes[P, Orderless];
KB1[pd_PD] := KB1[pd, {}, 1];
KB1[pd_PD, inside_, web_] := Module[
   {pos = First[Ordering[Length[Complement[List @@ #, inside]] & /@ pd]]},
   pd[[pos]] /. X[a_, b_, c_, d_] ⧴ KB1[
      Delete[pd, pos],
      Union[inside, {a, b, c, d}],
      Expand[
        web * (A P[a, d] P[b, c] + 1 / A P[a, b] P[c, d])] //.
       {P[e_, f_] P[f_, g_] ⧴ P[e, g], P[e_, _]^2 ⧴ P[e, e], P[e_, e_] → -A^2 - 1 / A^2}
      ]
   ];
KB1[PD[], _, web_] := Expand[web]
```

In[10]:=
```
TorusKnot[5, 4] // PD // KB1 // Timing
```

Out[10]= $\left\{0.046800, \frac{1}{A^{23}} + \frac{1}{A^{19}} + \frac{1}{A^{15}} + \frac{1}{A^{11}} + \frac{1}{A^{7}} - A - A^{5} - A^{9}\right\}$